RISC ErgoSum package available at https : // www3 . risc . jku . at/research/combinat/software/ergosum/

In[1]:=
```
<< RISC`GeneratingFunctions`
```

Package GeneratingFunctions version 0.9 written by Christian Mallinger
Copyright Research Institute for Symbolic Computation (RISC),
Johannes Kepler University, Linz, Austria

deriving three term recurrence for Hermite polynomials from closed form of exponential generating function

In[2]:=
```
defGFher = {f '[z] - 2 (x - z) f[z] == 0, f[0] == 1}
```

Out[2]=
```
{-2 (x - z) f[z] + f′[z] == 0, f[0] == 1}
```

In[3]:=
```
rec1 = DE2RE[defGFher, f[z], H[n]]
```

Out[3]=
```
{2 H[n] - 2 x H[1 + n] + (2 + n) H[2 + n] == 0, H[0] == 1, H[1] == 2 x}
```

In[4]:=
```
rec2 = REHadamard[rec1, {H[n + 1] - (n + 1) H[n] == 0, H[0] == 1}, H[n]]
```

Out[4]=
```
{2 (1 + n) H[n] - 2 x H[1 + n] + H[2 + n] == 0, H[0] == 1, H[1] == 2 x}
```

In[5]:=
```
ode2 = RE2DE[rec2, H[n], g[z]]
```

Out[5]=
```
{-1 - (-1 + 2 x z - 2 z^2) g[z] + 2 z^3 g′[z] == 0, g[0] == 1}
```

In[6]:=
```
? DifferentialRoot
```

Out[6]=

Symbol

DifferentialRoot[*lde*][*x*] gives the holonomic function h(x), specified by the linear differential equation *lde*[*h*, *x*].

DifferentialRoot[*lde*] represents a pure holonomic function h.

In[7]:=
```
TimeConstrained[DSolve[ode2, g[z], z], 20]
```

Out[7]=
```
$Aborted
```

In[8]:=
```
TimeConstrained[DSolve[{-1 - (-1 + 2 x z - 2 z^2) g[z] + 2 z^3 g′[z] == 0, g[0] == 1}, g[z], z], 20]
```

Out[8]=
```
$Aborted
```

Generating function of Chebyshev polynomials of the first kind

In[9]:=
```
recCheb1 = {T[n + 2] - 2 x T[n + 1] + T[n] == 0, T[0] == 1, T[1] == x}
```
Out[9]=
```
{T[n] - 2 x T[1 + n] + T[2 + n] == 0, T[0] == 1, T[1] == x}
```

In[10]:=
```
deCheb1 = RE2DE[recCheb1, T[n], F[z]]
```
Out[10]=
$$-1 + x z - (-1 + 2 x z - z^2) F[z] == 0$$

In[11]:=
```
Solve[deCheb1, F[z]]
```
Out[11]=
$$\left\{\left\{F[z] \to \frac{-1 + x z}{-1 + 2 x z - z^2}\right\}\right\}$$

derive a recurrence for $T_n(x) + (x + 1)^n$

In[12]:=
```
reMon = {p[n + 1] - (1 + x) p[n]  == 0, p[0] == 1}
```
Out[12]=
```
{-((1 + x) p[n]) + p[1 + n] == 0, p[0] == 1}
```

In[13]:=
```
re1 = recCheb1 /. {T → a}
re2 = reMon /. p → a
```
Out[13]=
```
{a[n] - 2 x a[1 + n] + a[2 + n] == 0, a[0] == 1, a[1] == x}
```
Out[14]=
```
{-((1 + x) a[n]) + a[1 + n] == 0, a[0] == 1}
```

In[15]:=
```
REPlus[re1, re2, a[n]]
```
Out[15]=
$$\{(-1 - x) a[n] + (1 + 2 x + 2 x^2) a[1 + n] + (-1 - 3 x) a[2 + n] + a[3 + n] == 0,$$
$$a[0] == 2, a[1] == 1 + 2 x, a[2] == 2 x + 3 x^2\}$$

In[16]:=
```
reLeg = {(n + 2) p[n + 2] - (2 n + 3) x p[n + 1] + (n + 1) p[n] == 0, p[0] == 1, p[1] == x}
```
Out[16]=
```
{(1 + n) p[n] - (3 + 2 n) x p[1 + n] + (2 + n) p[2 + n] == 0, p[0] == 1, p[1] == x}
```

Generating function for Legendre polynomials

In[17]:=
$$F(z) = \sum_{n=0}^{\infty} z^n P_n(x)$$
Out[17]=
$$\frac{1}{\sqrt{1 - 2 x z + z^2}}$$

In[18]:=
```
deLeg = RE2DE[reLeg, p[n], f[z]]
```
Out[18]=

$$\left\{-\left((x-z)\, f[z]\right)-\left(-1+2\,x\,z-z^2\right) f'[z] == 0,\ f[0] == 1\right\}$$

In[19]:=
```
DSolve[deLeg, f[z], z]
```
Out[19]=

$$\left\{\left\{f[z]\to \frac{1}{\sqrt{1-2\,x\,z+z^2}}\right\}\right\}$$

Christoffel - Darboux formula for Legendre polynomials $\sum_{k=}^{n}-(2k+1)P_k(1)\,P_k(x)$

In[20]:=
```
summand = REHadamard[reLeg, {(2 n + 1) p[n + 1] - (2 n + 3) p[n] == 0, p[0] == 1 / 2}, p[n]]
```
Out[20]=

$$\left\{(1+n)\,(5+2\,n)\,p[n]-(1+2\,n)\,(5+2\,n)\,x\,p[1+n]+(2+n)\,(1+2\,n)\,p[2+n] == 0,\ p[0] == \frac{1}{2},\ p[1] == \frac{3\,x}{2}\right\}$$

In[21]:=
```
sumCD = RECauchy[summand, {p[n] == 1}, p[n]]
```
Solve: Equations may not give solutions for all "solve" variables.

Out[21]=

$$\left\{(2+n)\,(7+2\,n)\,p[n]-(7+2\,n)\,(2+n+3\,x+2\,n\,x)\,p[1+n]+(3+2\,n)\,(3+n+7\,x+2\,n\,x)\,p[2+n]-(3+n)\,(3+2\,n)\,p[3+n] == 0,\ p[0] == \frac{1}{2},\ p[1] == \frac{1}{2}+\frac{3\,x}{2},\ p[2] == -\frac{3}{4}+\frac{3\,x}{2}+\frac{15\,x^2}{4}\right\}$$

In[22]:=
```
s[n_, x_] := Sum[(2 k + 1) / 2 LegendreP[k, x], {k, 0, n}]
```

In[23]:=
```
data = Table[Factor[s[nn, x]], {nn, 0, 30}];
```

In[24]:=
```
Take[data, 3]
```
Out[24]=

$$\left\{\frac{1}{2},\ \frac{1}{2}\,(1+3\,x),\ \frac{3}{4}\left(-1+2\,x+5\,x^2\right)\right\}$$

In[25]:=
```
GuessRE[data, p[n]]
```
Out[25]=

$$\left\{\left\{(2+n)\,(5+2\,n)\,p[n]+\left(-1-15\,x-16\,n\,x-4\,n^2\,x\right) p[1+n]+(2+n)\,(3+2\,n)\,p[2+n] == 0,\ p[0] == \frac{1}{2},\ p[1] == \frac{1}{2}\,(1+3\,x)\right\},\ ogf\right\}$$

In[26]:=
```
sumCD2 = %[[1]]
```
Out[26]=

$$\left\{(2+n)\,(5+2\,n)\,p[n]+\left(-1-15\,x-16\,n\,x-4\,n^2\,x\right) p[1+n]+(2+n)\,(3+2\,n)\,p[2+n] == 0,\ p[0] == \frac{1}{2},\ p[1] == \frac{1}{2}\,(1+3\,x)\right\}$$

In[27]:=
```
REPlus[sumCD, {(2 + n) (5 + 2 n) p[n] + (-1 - 15 x - 16 n x - 4 n^2 x) p[1 + n] + (2 + n) (3 + 2 n) p[2 + n] == 0,
   p[0] == -1/2, p[1] == -1/2 (1 + 3 x)}, p[n]]
```

Out[27]=
```
{-((2 + n) (7 + 2 n) p[n]) + (7 + 2 n) (2 + n + 3 x + 2 n x) p[1 + n] -
   (3 + 2 n) (3 + n + 7 x + 2 n x) p[2 + n] + (3 + n) (3 + 2 n) p[3 + n] == 0, p[0] == 0,
  p[1] == 1/2 (-1 - 3 x) + 1/2 (1 + 3 x), p[2] == 3/4 - (3 x)/2 - (15 x^2)/4 + 3/4 (-1 + 2 x + 5 x^2)}
```

In[28]:=
```
%[[All, 2]]
```

Out[28]=
```
{0, 0, 1/2 (-1 - 3 x) + 1/2 (1 + 3 x), 3/4 - (3 x)/2 - (15 x^2)/4 + 3/4 (-1 + 2 x + 5 x^2)}
```

In[29]:=
```
% // Factor
```

Out[29]=
```
{0, 0, 0, 0}
```

In[30]:=
```
REPlus[sumCD[[1]], sumCD2[[1]], p[n]]
```

Out[30]=
```
-((2 + n) (7 + 2 n) p[n]) + (7 + 2 n) (2 + n + 3 x + 2 n x) p[1 + n] -
  (3 + 2 n) (3 + n + 7 x + 2 n x) p[2 + n] + (3 + n) (3 + 2 n) p[3 + n] == 0
```

prove closed form for sumCD: $\frac{(n+\ )\left(P_{n+1}(x)-P_{n}(x)\right)}{2\,(x-1)}$

In[34]:=
```
MINUSreLeg = {(1 + n) p[n] - (3 + 2 n) x p[1 + n] + (2 + n) p[2 + n] == 0, p[0] == -1, p[1] == -x}
```

Out[34]=
```
{(1 + n) p[n] - (3 + 2 n) x p[1 + n] + (2 + n) p[2 + n] == 0, p[0] == -1, p[1] == -x}
```

compute a recurrence for the closed form:

In[35]:=
```
reCF = REHadamard[{(n + 1) p[n + 1] - (n + 2) p[n] == 0, p[0] == 1 / (2 (x - 1))},
   REPlus[RESubsequence[reLeg, p[n], n + 1], MINUSreLeg, p[n]], p[n]]
```

Out[35]=
```
{(2 + n) (3 + n) p[n] - 2 (3 + n) (5 + 2 n) x p[1 + n] + (19 + 12 n + 2 n^2 + 35 x^2 + 24 n x^2 + 4 n^2 x^2) p[2 + n] -
   2 (3 + n) (7 + 2 n) x p[3 + n] + (3 + n) (4 + n) p[4 + n] == 0, p[0] == 1/2,
  p[1] == (-1 - 2 x + 3 x^2)/(2 (-1 + x)), p[2] == (3 (1 - 3 x - 3 x^2 + 5 x^3))/(4 (-1 + x)), p[3] == (3 + 12 x - 30 x^2 - 20 x^3 + 35 x^4)/(4 (-1 + x))}
```

simplify the initial conditions for easier comparison:

In[37]:=
```
Factor /@ Rest[reCF][[All, 2]]
```

Out[37]=

$$\left\{\frac{1}{2}, \frac{1}{2}(1+3x), \frac{3}{4}\left(-1+2x+5x^2\right), \frac{1}{4}\left(-3-15x+15x^2+35x^3\right)\right\}$$

In[36]:=
```
sumCD2
```

Out[36]=

$$\left\{(2+n)(5+2n)\,p[n]+\left(-1-15x-16nx-4n^2x\right)p[1+n]+(2+n)(3+2n)\,p[2+n]==0,\ p[0]==\frac{1}{2},\ p[1]==\frac{1}{2}(1+3x)\right\}$$

In[39]:=
```
REPlus[sumCD2[[1]], reCF[[1]], p[n]]
```

Out[39]=

$$(2+n)(3+n)\,p[n]-2(3+n)(5+2n)\,x\,p[1+n]+\left(19+12n+2n^2+35x^2+24nx^2+4n^2x^2\right)p[2+n]-2(3+n)(7+2n)\,x\,p[3+n]+(3+n)(4+n)\,p[4+n]==0$$

same recurrence as for the closed form